Appendix A. Listing of computer code to generate axisymmetric hull

```
C
C ****************************************************************
C
C                    D A R P A 2 G E N . F O R
C
C ****************************************************************
C
C           THIS PROGRAM CONTAINS FOLLOWING EQUATIONS FOR
C           GENERATING OFFSETS IN FEET FOR DARPA2 MODEL
C           WITH (FULL/MODEL) SCALE RATIO = 24.
C
C      INCLUDED ARE:
C        BOW EQ.                FOR        0.0 FT <= X <=  3.333333 FT,
C        PARALLEL MID-BODY EQ.  FOR   3.333333 FT <= X <= 10.645833 FT,
C        AFTERBODY EQ.          FOR  10.645833 FT <= X <= 13.979167 FT,
C        AFTERBODY CAP EQ.      FOR  13.979167 FT <= X <= 14.291667 FT.
C
C   AS SET UP HERE, OFFSETS ARE COMPUTED EVERY 0.1 FT.
C   (EXCEPT IN FIRST 0.5 FT, WHERE THEY ARE EVERY 0.01 FT)
C
C
      DIMENSION X(300), Y(300)
      REAL K0, K1
C
C ********************************
C
C        DEFINE CONSTANTS
C
C ********************************
C
      RMAX = 0.8333333
      XB =  3.333333
      XM = 10.645833
      XA = 13.979167
      XC = 14.291667
      CB1 = 1.126395101
      CB2 = 0.442874707
      CB3 = 1.0/2.1
      RH = 0.1175
      K0 = 10.0
      K1 = 44.6244
C
      XX = -0.01
      DX = 0.01
      DO 1000 I=1,300
      NP = I
      XX = XX + DX
      IF(XX.GE.0.5) DX = 0.1
      IF(XX.GE.XA) DX = 0.01
      IF(XX.GE.XB) GO TO 200
C
C ********************************
C
C         BOW EQUATION
C
C ********************************
C
      A = 0.3*XX - 1.0
      A3 = A**3
      A4 = A**4
```

```
      B = 1.2*XX + 1.0
      R = CB1*XX*A4 + CB2*XX*XX*A3 + 1.0 - A4*B
      R = RMAX*(R**CB3)
      X(I) = XX
      Y(I) = R
      GO TO 1000
C
  200 CONTINUE
      IF(XX.GE.XM) GO TO 400
C
C *********************************
C
C   PARALLEL MID-BODY EQUATION
C
C *********************************
C
      X(I) = XX
      Y(I) = RMAX
      GO TO 1000
C
  400 CONTINUE
      IF(XX.GE.XA) GO TO 600
C
C *********************************
C
C       AFTERBODY EQUATION
C
C *********************************
C
      XI = (13.979167 - XX)/3.333333
      C1 =                 RH*RH
      C2 =                                RH*K0                    *XI*XI
      C3 = ( 20.0 - 20.0*RH*RH - 4.0*RH*K0 - 0.333333*K1)*XI**3
      C4 = (-45.0 + 45.0*RH*RH + 6.0*RH*K0 +           K1)*XI**4
      C5 = ( 36.0 - 36.0*RH*RH - 4.0*RH*K0 -           K1)*XI**5
      C6 = (-10.0 + 10.0*RH*RH +     RH*K0 + 0.333333*K1)*XI**6
      R = RMAX*(C1+C2+C3+C4+C5+C6)**0.5
      X(I) = XX
      Y(I) = R
      GO TO 1000
C
  600 CONTINUE
      IF(XX.GE.XC) GO TO 1100
C
C *********************************
C
C     AFTERBODY CAP EQUATION
C
C *********************************
C
      R = 1.0 - (3.2*XX - 44.733333)**2
      R = RH*RMAX*(R**0.5)
      X(I) = XX
      Y(I) = R
 1000 CONTINUE
 1100 CONTINUE
      X(NP) = XC
      Y(NP) = 0.0
C
C *********************************
```

```
C
C     WRITE OFFSETS TO TAPE 6
C     IN IPLOT FORMAT
C
C ********************************
C
      WRITE(6,1)
    1 FORMAT('DARPA2')
      WRITE(6,2)
    2 FORMAT('MODEL WITH (MODEL/FULL) = 24')
      WRITE(6,3) NP
    3 FORMAT(I5)
      WRITE(6,4) (X(I),Y(I),I=1,NP)
    4 FORMAT(2F10.5,3X,2F10.5,3X,2F10.5)
C
C  ALL DONE, PROGRAM ENDS
C
      STOP
      END
```

APPENDIX B

LISTING OF COMPUTER CODE TO GENERATE FAIRWATER

Appendix B. Listing of computer code to generate fairwater

```
C
C  REVISED 11-JANUARY-1989
C
C*****************************************************************
C
C           D A R P A 2 G E N 2 . F O R
C
C*****************************************************************
C
C  THIS PROGRAM CONTAINS FOLLOWING EQUATIONS FOR
C  GENERATING OFFSETS IN FEET FOR THE SAIL OF
C  THE DARPA2 MODEL WITH (FULL/MODEL) SCALE
C  RATIO = 24.
C
C  INCLUDED ARE:
C  SAIL FOREBODY EQ.   FOR   3.032986 FT <= X <=  3.358507 FT
C                            0.833333 FT <= Z <=  1.507813 FT
C  SAIL MID-BODY EQ.   FOR   3.358507 FT <= X <=  3.559028 FT
C                            0.833333 FT <= Z <=  1.507813 FT
C  SAIL AFTERBODY EQ.  FOR   3.559028 FT <= X <=  4.241319 FT
C                            0.833333 FT <= Z <=  1.507813 FT
C  SAIL CAP EQ.        FOR   3.032986 FT <= X <=  4.241319 FT
C                            1.507813 FT <= Z <=  1.562501 FT
C
C  OFFSETS ARE COMPUTED EVERY .005 FT.
C
C
      DIMENSION
     .       NP(300),
     .       X(300,50,3)
C
C*********************************
C
C  DEFINE CONSTANTS
C
C*********************************
C
      A1     = 2.094759
      B1     = 0.207178
      A3     = 2.908891
      B3     = 1.234491
      C3     = 3.444817
      D3     = 3.850435
      E3     = 2.080019
      HMAX   = 0.109375
      DX     = 0.005
      DX0    = 0.005
      XXCST  = 3.032986
      XXAFN  = 4.241319
      XXFFN  = 3.358507
      XXMFN  = 3.559028
      XZST   = 1.507813
C
C*********************************
C
C  CALCULATE
C
C*********************************
C
            XX=XXCST-DX
      DO 1000 I = 1,300
      XZ=XZST
      X(I,1,3)=XZ
```

```
      J=1
      XX=XX+DX
      X(I,1,1)=XX
      IF (XX .GT. XXAFN) THEN
                NI=I-1
                GOTO 1014
      ENDIF
      IF (XX .GT. XXFFN) GOTO 1002
C
C*********************************
C
C   SAIL FOREBODY EQUATION
C
C*********************************
C
      D=3.072000*(XX-3.032986)
      DM1=D-1
      A=2*D*(DM1**4)
      B=D*D*(DM1**3)/3
      C=1-((DM1**4)*(4*D+1))
      X(I,1,2)=HMAX*(SQRT(A1*A+B1*B+C))
      GOTO 1004
C
C*********************************
C
C   SAIL MID-BODY EQUATION
C
C*********************************
C
 1002         CONTINUE
      IF (XX .GT. XXMFN) GOTO 1003
      X(I,1,2)=HMAX
      GOTO 1004
C
C*********************************
C
C   SAIL AFTER BODY EQUATION
C
C*********************************
C
 1003         CONTINUE
      E=(4.241319-XX)/.6822917
      F=E-1
      G=2.238361*E*F**4
      H=3.106529*(E**2)*(F**3)
      P=1-(F**4)*(4*E+1)
      X(I,1,2)=.1093750*(G+H+P)
C
C*********************************
C
C   SAIL CAP EQUATION
C
C*********************************
C
 1004         CONTINUE
      XZEND=(X(I,1,2)/2)+1.507813
      NP(I)=1
      DO 1008 J=2,50
      ICON1=0
 1005         XZ=XZ+DX
      X(I,J,3)=XZ
      IF (XZ .GT. XZEND) THEN
                ICON1=ICON1+1
```

```
            IF (ICON1 .EQ. 1) THEN
                              XZ=XZ-DX
                    DX=.0005
                    GOTO 1005
            ENDIF
            IF (ICON1 .EQ. 2) THEN
                    X(I,J,2)=0.0
                    X(I,J,3)=XZEND
                    NP(I)=J
                    ICON1=0
                    DX=DX0
                    GOTO 1000
            ENDIF
    ENDIF
    ADUM=(X(I,1,2)**2)-((2*(XZ-XZST))**2)
    X(I,J,2)=SQRT(ADUM)
 1008       CONTINUE
 1000       CONTINUE
C
C*********************************
C
C   WRITE OFFSETS TO TAPE6
C     IN IPLOT FORMAT
C
C*********************************
C
 1014       OPEN(6,STATUS='NEW',FORM='FORMATTED',FILE='TP6')
    WRITE(6,1015)
 1015       FORMAT('DARPA2 SAIL')
    WRITE(6,1016)
 1016       FORMAT('MODEL WITH (MODEL/FULL) = 24')
    WRITE(6,1017)NI
 1017       FORMAT(I5)
    WRITE(6,1018) (X(I,1,1),X(I,1,2),I=1,NI)
 1018       FORMAT(2F10.5,3X,2F10.5,3X,2F10.5)
    DO 1013 I=1,NI,8
    WRITE(6,1009)I
 1009       FORMAT(I3)
    WRITE(6,1010)X(I,1,1)
 1010       FORMAT(' X=' ,F7.3,' FEET')
    WRITE(6,1011) (NP(I)+1)
 1011       FORMAT(I5)
    WRITE(6,1012) X(I,1,2),1.5
    WRITE(6,1012) (X(I,J,2),X(I,J,3),J=1,NP(I))
 1012       FORMAT(2F10.5,3X,2F10.5,3X,2F10.5)
 1013       CONTINUE
  666       STOP
    END
```

APPENDIX C

LISTING OF COMPUTER CODE TO GENERATE STERN APPENDAGES

Appendix C. Listing of computer code to generate stern appendages

```
C
C ****************************************************************
C
C           D A R P A 2 S T E R N A P P . F O R
C
C ****************************************************************
C
C
C         THIS PROGRAM DEFINES THREE-DIMENSIONAL (X, Y, Z)
C         OFFSETS FOR DARPA2 STERN APPENDAGES WITH TRAILING
C         EDGE LOCATED AT THREE DIFFERENT VALUES OF AXIAL
C         LENGTH X.  FOR EACH AXIAL POSITION, FOUR IDENTICAL
C         STERN APPENDAGES ARE MOUNTED ON THE AXISYMMETRIC
C         HULL SURFACE AT TOP-DEAD-CENTER, 90 DEG, 180 DEG,
C         AND 270 DEG AZIMUTHALLY.
C
C
C     (X, RR, Z) = CARTESIAN COORDINATES IN FEET
C
C     H          = X COORDINATE OF STERN APPENDAGE TRAILING EDGE.
C                    H(1) = 12.729617
C                    H(2) = 13.146284 = BASELINE
C                    H(3) = 13.562950
C
C     CY         = CHORD LENGTH = -0.466308*RR + 0.88859
C
      DIMENSION XXI(19), H(3)
C
      PARAMETER RH = 0.1175, AK0 = 10.0, AK1 = 44.6244
      PARAMETER NP = 19, RMAX = 0.833333
C
      DATA XXI/0.0,   0.005, 0.0125, 0.025, 0.050, 0.075, 0.100,
     1         0.150, 0.200, 0.2500, 0.300, 0.400, 0.500, 0.600,
     2         0.700, 0.800, 0.9000, 0.950, 1.000/
C
      DATA H/12.729617, 13.146284, 13.562950/
C
C ******************************************
C
C     LOOP ON THE LOCATION OF STERN
C       APPENDAGE TRAILING EDGE
C
C ******************************************
C
      DO 900 K=1,3
      HH = H(K)
      WRITE(6,1) HH
    1 FORMAT(//2X,'STERN APPENDAGE TRAILING EDGE LOCATED AT X ='
     1         F10.5)
      DX = 0.05
      X = HH + DX
C
C ******************************************
C
C       LOOP ON THE AXIAL POSITION X.
C     BEGIN AT STERN APPENDAGE TRAILING
C        EDGE AND MOVE FORWARD IN X.
C
C ******************************************
C
```

```
      DO 800 J = 1,32
      X = X - DX
      IF(X.GT.HH) GO TO 800
C
C *********************************************
C
C     DE INE HULL RADIUS AT VALUE OF X
C
C *********************************************
C
      XIB = (13.979167 - X)/3.333333
      A =                  RH*RH +     RH*AK0                    *XIB*XIB
      B = ( 20.0 - 20.0*RH*RH - 4.0*RH*AK0 - 0.333333*AK1)*XIB**3
      C = (-45.0 + 45.0*RH*RH + 6.0*RH*AK0 +          AK1)*XIB**4
      D = ( 36.0 - 36.0*RH*RH - 4.0*RH*AK0 -          AK1)*XIB**5
      E = (-10.0 + 10.0*RH*RH +     RH*AK0 + 0.333333*AK1)*XIB**6
      RHA = A + B + C + D + E
      RHA = RMAX*SQRT(RHA)
      RHAS = RHA*RHA
      RR = 0.075
      DELR = 0.025
      ITR = 0
C
C *********************************************
C
C              LOOP ON RADIUS.
C            BEGIN WITH R = 0.1
C
C *********************************************
C
      DO 700 I=1,31
      RR = RR + DELR
  620 CONTINUE
      CY = -0.466308*RR + 0.88859
      XI = (X-HH)/CY + 1.0
      IF(XI.LT.0.0 .OR. XI.GT.1.0) GO TO 700
C
C *********************************************
C
C  DEFINE STERN APPENDAGE
C
C *********************************************
      Z = 0.29690*SQRT(XI) - 0.12600*XI - 0.35160*XI*XI
     1    + 0.28520*XI**3 - 0.10450*XI**4
      Z = CY * Z
      SRS = RR*RR + Z*Z
C
C *********************************************
C
C        IF STERN APPENDAGE LOCATED
C         INSIDE BODY, INCREASE R
C
C *********************************************
C
      IF(SRS.LT.RHAS. AND. ITR.EQ.0) GO TO 700
C
C *********************************************
C
C          IF STERN APPENDAGE LOCATED
C              ON BODY SURFACE,
```

```
C           GO TO 710 TO DEFINE
C          STERN APPENDAGE SECTION.
C
C ******************************************
C
      IF(ABS(SRS-RHAS).LE.0.00001) GO TO 710
C
C ******************************************
C
C         STERN APPENDAGE IS "CLOSE"
C         TO HULL RADIUS, GET CLOSER.
C
C ******************************************
C
      ITR = ITR + 1
      IF(ITR.GT.20) STOP1
      DELR = 0.5*DELR
      IF(SRS.GT.RHAS) RR = RR - DELR
      IF(SRS.LE.RHAS) RR = RR + DELR
      GO TO 620
  700 CONTINUE
      GO TO 800
  710 CONTINUE
C
C ******************************************
C
C     SOLVE FOR STERN APPENDAGE SECTION
C             AT GIVEN RADIUS
C
C ******************************************
C
      CY = -0.466308*RR + 0.88859
      I750 = 0
      XINIT = (X-HH)/CY + 1.0
C
C ******************************************
C
C                 LOOP ON XI
C
C ******************************************
C
      DO 750 I=1,NP
      XI = XXI(I)
      IF(XI.LT.XINIT) GO TO 750
  740 CONTINUE
      XI = XXI(I)
      IF(I750.EQ.0) XI = XINIT
      XXX = (XI-1.0)*CY + HH
      IF(XI.LT.0.0 .OR. XI.GT.1.0) GO TO 750
      Z = 0.29690*SQRT(XI) - 0.12600*XI - 0.35160*XI*XI
     1    + 0.28520*XI**3 - 0.10450*XI**4
      Z = CY * Z
C
C ******************************************
C
C         PRINT X, Y, (+/-)Z VALUES
C            TO PRINTER FILE 6
C
C ******************************************
C
```

```
      IF(I750.EQ.0) WRITE(6,2)
    2 FORMAT(/6X,1HX,9X,1HY,6X,6H(+/-)Z)
      WRITE(6,3) XXX, RR, Z
    3 FORMAT(3F10.5)
      I750 = I750 + 1
      RBSMAX = RR
      IF(I750.EQ.1) GO TO 740
  750 CONTINUE
  800 CONTINUE
C
C ********************************************
C
C    COMPUTED ALL STERN APPENDAGE SECTIONS
C            WHICH INTERSECT HULL.
C         NOW COMPUTE STERN SECTIONS
C     WITH RADIUS LARGER THAN HULL RADIUS.
C
C ********************************************
C
      DELR = 0.05
      DO 850 I=1,NP
      RO = RR
      RR = RBSMAX + I*DELR
      IF(RR.GT.RMAX) RR = RMAX
      IF(RR.EQ.RO) GO TO 900
      CY = -0.466308*RR + 0.88859
      WRITE(6,2)
      DO 840 J=1,NP
      XI = XXI(J)
      XXX = (XI-1.0)*CY+HH
      Z = 0.29690*SQRT(XI) - 0.12600*XI - 0.35160*XI*XI
     1    + 0.28520*XI**3 - 0.10450*XI**4
      Z = CY * Z
C
C ********************************************
C
C          PRINT X, Y, (+/-)Z VALUES
C             TO PRINTER FILE 6
C
C ********************************************
C
      WRITE(6,3) XXX, RR, Z
  840 CONTINUE
  850 CONTINUE
  900 CONTINUE
      STOP
      END
```

APPENDIX D

LISTING OF COMPUTER CODE TO GENERATE RING WINGS

Appendix D. Listing of computer code to generate ring wings

```
C
C ***********************************************************
C
C       P R O G R A M       D A R P A 2 W I N G S
C
C ***********************************************************
C
C    THIS PROGRAM DEFINES THE DARPA2 RING WINGS
C
C    THE DARPA2 WINGS USE THE NACA66 (DTNSRDC MOD)
C               THICKNESS DISTRIBUTION
C                     AND
C              THE NACA A=0.4 MEANLINE
C
      DIMENSION XC(26), YC(26), YCP(26)
      DIMENSION B(17), YT(26)
      DIMENSION XU(26), YU(26), XL(26), YL(26)
      DIMENSION XDLE(2), YDLE(2), XDTE(2), YDTE(2)
C
C  XC ARRAY ARE THE X/C VALUES CURRENTLY USED TO DEFINE WING.
C
      DATA XC/0.0, 0.005, 0.0075, 0.0125, 0.025, 0.05, 0.075, 0.10,
     1        0.15, 0.20, 0.25, 0.30, 0.35, 0.40, 0.45, 0.50, 0.55,
     2        0.60, 0.65, 0.70, 0.75, 0.80, 0.85, 0.90, 0.95, 1.0/
C
C  B ARRAY CONTAINS COEFFICIENTS FOR CALCULATION OF THICKNESS DISTR.
C
      DATA B/0.43756, -0.08136, -0.06496, -0.01926, -0.00185,
     1       0.00348,  0.00156, -0.00113, -0.00058,  0.00027,
     2       0.00080,  0.00006, -0.00027, -0.00033,  0.00005,
     3       0.00014,  0.00008/
C
C  XDLE, YDLE ARE LEADING EDGE X, R OF WING
C  XDTE, YDTE ARE TRAILING EDGE X, R OF WING
C
      DATA XDLE/13.46990, 13.46990/
      DATA YDLE/0.43004, 0.47681/
      DATA XDTE/14.21661,14.2074/
      DATA YDTE/0.35659, 0.33856/
C
C  THE ENTIRE PROGRAM IS EXERCISED TWO TIMES.
C    THE FIRST TIME, WING 1 IS DEFINED.
C       WING 1 HAS LEADING EDGE AT (X=13.46990, Y=0.43004)
C              AND TRAILING EDGE AT (X=14.23, Y=0.3558)
C    THE SECOND TIME, WING 2 IS DEFINED.
C       WING 2 HAS LEADING EDGE AT (X=13.46990, Y=0.3558)
C              AND TRAILING EDGE AT (X=14.23, Y=0.33628)
C
      NXC = 26
      DO 1000 KK = 1,2
      WRITE(6,2)
C
C  ********************
C
C   DEFINE MEAN LINE
C
```

```
C  ********************
C
      DO 100 I=1,NXC
      X = XC(I)
      D = 0.4 - X
      E = 1.0 - X
      IF(ABS(X-0.0).LE.1.0E-20) X = 1.0E-30
      IF(ABS(D).LE.1.0E-20) D = 1.0E-30
      IF(ABS(E).LE.1.0E-20) E = 1.0E-30
      YC(I) = -0.049921*(0.5*D*D*ALOG(ABS(D)) - 0.5*E*E*ALOG(E)
     1             + 0.25*E*E - 0.25*D*D)
      YC(I) = YC(I) +0.029953*(X*ALOG(X) + 0.227828 - 0.531076*X)
C
      YCP(I) =  -0.049921*(E*ALOG(E) - D*ALOG(ABS(D)))
     1         + 0.02995253*(ALOG(X) + 0.4689244)
C        WRITE(6,1) I, 100.*XC(I), 100.*YC(I), YCP(I)
    1   FORMAT(I5,2F10.3,F10.5)
  100 CONTINUE
C      WRITE(6,2)
    2 FORMAT(//)
C
C  *********************************
C
C    DEFINE THICKNESS DISTRIBUTION
C
C  *********************************
C
      NSER = 17
      DO 200 I=1,NXC
        X = XC(I)
        IF(I.GE.16) GO TO 150
        OM = ACOS(2.0*X-1.0)
        YY = 0.0
        DO 125 J=1,NSER
           YY = YY + B(J)*SIN(J*OM)
  125   CONTINUE
        YT(I) = YY
        GO TO 199
  150   CONTINUE
        XC1 = 1.0-XC(I)
        YT(I) =  0.033333 + 1.696969*XC1 - 1.441945*XC1*XC1
     1          -0.366363*XC1**3 + 0.333049*XC1**4
  199   CONTINUE
        YT(I) = 0.1*YT(I)
C        WRITE(6,3) I,X,YT(I)
    3   FORMAT(I5,F10.3,F10.5)
  200 CONTINUE
C      WRITE(6,2)
C
C ***************************
C
C    DEFINE DARPA2 WING
C
C ***************************
C
      XLINIT = 0.9425
      YLINIT = 0.0258
      CHORD = 0.0525
      XU(1) = 0.0
      YU(1) = 0.0
      XL(1) = 0.0
      YL(1) = 0.0
      I = 1
```

```
C      WRITE(6,6)
    6 FORMAT(2X,1HI,3X,4H XU ,5X,4H YU ,5X,4H XL ,5X,4H YL ,6X,3HX/C,
     1       7X,2HYT,7X,2HYC,4X,7HDYC/DXC/)
    5 FORMAT(I3, 4F9.5, F9.4,4F9.5)
C        WRITE(6,5) I,XU(I),YU(I),XL(I),YL(I),XC(I),YT(I),YC(I),YCP(I)
      DO 300 I=2,NXC
        TH = ATAN(YCP(I))
        SINTH = SIN(TH)
        COSTH = COS(TH)
        XU(I) = XC(I) - YT(I)*SINTH
        YU(I) = YC(I) + YT(I)*COSTH
        XL(I) = XC(I) + YT(I)*SINTH
        YL(I) = YC(I) - YT(I)*COSTH
C        WRITE(6,5) I,XU(I),YU(I),XL(I),YL(I),XC(I),YT(I),YC(I),YCP(I)
  300 CONTINUE
C
C ***********************************
C
C  DEFINE PHYSICAL WING DIMENSIONS
C
C ***********************************
C
      PHI = ATAN2((YDTE(KK)-YDLE(KK)),(XDTE(KK)-XDLE(KK)))
      CS = COS(PHI)
      SN = SIN(PHI)
      CHORD = SQRT((YDTE(KK)-YDLE(KK))**2 + (XDTE(KK)-XDLE(KK))**2)
C      WRITE(6,444) XDLE(KK),YDLE(KK), XDTE(KK),YDTE(KK)
  444 FORMAT(2X,'(XDLE,YDLE) = ',F10.5/2X,'(XDTE,YDTE) = ',F10.5)
      WRITE(6,6)
      DO 400 I=1,NXC
      XUU = XU(I)
      XU(I) = XDLE(KK) + CHORD*(XU(I)*CS - YU(I)*SN)
      YU(I) = YDLE(KK) + CHORD*(XUU*SN + YU(I)*CS)
      XLL = XL(I)
      XL(I) = XDLE(KK) + CHORD*(XL(I)*CS - YL(I)*SN)
      YL(I) = YDLE(KK) + CHORD*(XLL*SN + YL(I)*CS)
        WRITE(6,5) I,XU(I),YU(I),XL(I),YL(I),XC(I),YT(I),YC(I),YCP(I)
    4   FORMAT(I5,4F10.5)
  400 CONTINUE
C
C ******************************************
C
C   WRITE WING OFFSETS TO FILE 7 FOR IPLOT
C
C ******************************************
C
      IF(KK.EQ.1) WRITE(7,10)
   10 FORMAT('S1')
      IF(KK.EQ.2) WRITE(7,11)
   11 FORMAT('S2')
      IF(KK.EQ.1) WRITE(7,12)
   12 FORMAT('DARPA2 RING WING 1 ')
      IF(KK.EQ.2) WRITE(7,15)
   15 FORMAT('DARPA2 RING WING 2 ')
      NXC2 = 2*NXC
      WRITE(7,13) NXC2
   13 FORMAT(I5)
      DO 500 I=1,NXC2
        IF(I.GT.NXC) GO TO 450
        WRITE(7,14) XU(I),YU(I)
        GO TO 500
  450   CONTINUE
        J = NXC2-I+1
```

```
        GO TO 500
  450   CONTINUE
        J = NXC2-I+1
        WRITE(7,14) XL(J),YL(J)
  500 CONTINUE
   14 FORMAT(2F10.5)
C
C ***********************************
C
C  PRINT OFFSETS IN AMI FORMAT
C        ONTO FILE 9.
C
C ***********************************
C
      DO 600 I=1,NXC2
      IF(I.GT.NXC) GO TO 550
      J = NXC-I+1
      WRITE(9,14) XL(J), YL(J)
      GO TO 600
  550 CONTINUE
      K = I-NXC
      WRITE(9,14) XU(K), YU(K)
  600 CONTINUE
 1000 CONTINUE
      STOP
      END
```

APPENDIX E

LISTING OF COMPUTER CODE TO GENERATE RING WING STRUTS

Appendix E. Listing of computer code to generate ring wing struts

```
C
C ********************************************
C
C    P R O G R A M    D A R P A 2 S T R U T
C
C ********************************************
C
C
C   THIS PROGRAM DEFINES THE STRUT WHICH ATTACHES THE
C    DARPA2 AXISYMMETRIC HULL TO THE DARPA2 RING WINGS.
C    THE SAME BASIC STRUT IS USED TO ATTACH BOTH
C    RING WING 1 AND RING WING 2.  THE UPPER PORTION OF
C    THE STRUT MUST BE MODIFIED TO FIT EACH WING.
C
C   THIS PROGRAM DEFINES A SINGLE STRUT WHICH WOULD
C    ATTACH TO THE DARPA2 AXISYMMETRIC HULL ALONG
C    THE UPPER SURFACE (I.E., THE SURFACE WITH THE
C    FAIRWATER).  FOUR IDENTICAL AXIMUTHALLY EQUALLY-
C    SPACED STRUTS WILL ATTACH THE RING WING AND THE HULL
C    AT A 45 DEGREE INCREMENT FROM THE SURFACE PRESSURE
C    TAP LOCATIONS.  THE STRUTS WILL BE PLACED
C    AT 90 DEGREE INCREMENTS.
C
C   THE BASIC STRUT SHAPE IS A NACA 0012 THICKNESS
C    DISTRIBUTION MODIFIED TO END AT A POINT.
C    THE CHORD LENGTH IS 0.25 FEET.
C
C  STRUT LEADING EDGE ATTACHES TO:
C       HULL         AT X=13.589,   R=0.14726
C       RING WING 1 AT X=13.63845, R=0.36886    (X/C=0.233)
C       RING WING 2 AT X=13.64487, R = 0.39755  (X/C=0.25)
C
C  STRUT TRAILING EDGE ATTACHES TO:
C       HULL         AT X=13.83582, R=0.10547
C       RING WING 1 AT X=13.88818, R=0.34002    (X/C=0.5651)
C       RING WING 2 AT X=13.89023, R=0.34932    (X/C=0.5804)
C
      DIMENSION XC(19), XUL(19), YUL(19), ZU(19), ZL(19)
C
      DATA XC/0., 0.005, 0.0125, 0.025, 0.05, 0.075, 0.1, 0.15,
     1        0.2, 0.25, 0.3, 0.4, 0.5, 0.6, 0.7, 0.8, 0.9, 0.95, 1.0/
C
      NP = 19
      NR = 10
C
C  R1 = LE RADIUS OF STRUT AT HULL ATTACHMENT        = 0.14726
C  R2 = LE RADIUS OF STRUT AT RING WING 2 ATTACHMENT = 0.39755
C
      R1 = 0.14726
      R2 = 0.39755
      DELR = (R2-R1)/(NR-1)
      R = R1 - DELR
C
```

```
    2 FORMAT(/8X,'STRUT OFFSETS (IN FEET) AT RADIUS = ',F10.5/
     1     3X,'X PORT '3X,'Y PORT ',3X,'Z PORT ',5X,'X STBD ',3X,
     2     'Y STBD ',3X,'Z STBD '/)
      X0 = 0.223221*R + 13.556128
C
C  100 LOOP OVER STRUT CHORD
C
      DO 100 I=1,NP
      XI = XC(I)
      XUL(I) = X0 + 0.243995*XI
      YUL(I) = R  - 0.054465*XI
      ZT = 0.15*(0.29690*SQRT(XI) - 0.12600*XI - 0.35160*XI*XI
     1           + 0.28520*XI**3 - 0.10450*XI**4)
      ZU(I) = ZT
      ZL(I) = -ZT
      WRITE(6,3) XUL(I),YUL(I),ZU(I),XUL(I),YUL(I),ZL(I)
    3 FORMAT(3F10.5,2X,3F10.5)
  100 CONTINUE
  200 CONTINUE
      STOP
      END
```